NOTES SCIENTIFIQUES et TECHNIQUES
DU BUREAU des LONGITUDES

S025

SOLUTION D'UN PROBLEME DES N CORPS DEVELOPPEE EN SERIES DE TAYLOR D'ORDRES ELEVES. APPLICATION AU SYSTEME SOLAIRE

Claude Le Guyader

Service des Calculs et de Mécanique Céleste
du Bureau des Longitudes

UA 707
77 Avenue Denfert Rochereau
75014 Paris

mars 1990

ABSTRACT

In this paper, we use a method similar to the Broucke's one (1971), for the numerical integration of the N body problem.
Remarking that in the differential equations, all the functions of the second members possess the same form,we construct formal recurrent formulae in order to derive only one of them,and then we obtain the solution expanded in Taylor's series.
To test this method, we apply it to the system of the nine principal planets in their relative motions around the Sun.
Then, in order to estimate the accuracy of this method, we have integrated this system over 40000 days, forwards and reverse in time, with a step of 4 days.
Finally we compare the results obtained with the ephemeris DE200 of JPL(STANDISH,1982), to which the relativistic perturbations and those due to the Moon and the small planets were first substracted.

NOTATIONS GENERALES

Vecteurs Dérivées : $\vec{r}^{(k)} = (x^{(k)}, y^{(k)}, z^{(k)})$ k entier ≥ 0

Vecteur Position : $\vec{r} = (x, y, z)$ ou $\vec{r}^{(0)} = (x^{(0)}, y^{(0)}, z^{(0)})$

Vecteur Vitesse : $\dot{\vec{r}} = (\dot{x}, \dot{y}, \dot{z})$ ou $\vec{r}^{(1)} = (x^{(1)}, y^{(1)}, z^{(1)})$

Distance : $R = (x^2 + y^2 + z^2)^{\frac{1}{2}}$

$$F = -\frac{1}{R}$$

$$F_{\vec{r}} = \frac{\vec{r}}{R^3}$$ soit :

$$F_x = \frac{x}{R^3} \; ; \qquad F_y = \frac{y}{R^3} \; ; \qquad F_z = \frac{z}{R^3} \; ;$$

H = constante de l' énergie

K = carré de la constante de Gauss

$$C_k^h = \frac{k!}{h!\,(k-h)!}$$ h entier ≥ 0

INTRODUCTION

En dehors des méthodes, de type Gauss Jackson ou Runge Kutta par exemples, habituellement employées pour résoudre numériquement des problèmes de mécanique céleste, on peut aussi lorsque les mouvements étudiés restent stables, en rechercher des solutions développées en séries de Taylor d'ordres élevés.
De tels développements, en agrandissant les pas d'intégration diminuent alors les erreurs d'arrondi.

Principe de la Méthode utilisée

Pour obtenir ces solutions nous utiliserons dans ce qui suit une méthode analogue à celle décrite par BROUCKE(1971).
Celle ci consiste pour commencer à construire des formules de récurrence donnant formellement les dérivées successives de fonctions telles que :

$$F_{\vec{r}} = \frac{\vec{r}}{R^3}$$

puis à généraliser leurs utilisations à toutes les fonctions composant les second membres des équations différentielles d'un problème des N Corps.
On obtient alors ensuite de proche en proche, les dérivées $\vec{r}_i^{(k)}$ nécessaires aux développements recherchés.

Application au Système Solaire et Résultats

Nous allons, de cette manière, étudier le système des neuf planètes principales dans leurs mouvements relatifs autour du Soleil.
Nous utiliserons pour cela des formules de récurrence semblables à celles de Broucke(1971), puis à l'aide de la solution en séries de Taylor obtenue nous construirons une éphéméride du système solaire sur un intervalle de temps de 40000 jours.

Nous en donnerons ensuite les principaux résultats en comparaison avec une éphéméride de référence, celle DE200 du JPL(STANDISH,1982) à laquelle on aura préalablement soustrait les perturbations relativistes et celles dues à la Lune et aux petites planètes(LESTRADE et BRETAGNON,1982).
Nous améliorerons enfin notre éphéméride en réintégrant plusieurs fois le système étudié,mais en corrigeant légèrement à chaque fois les conditions initiales d'une nouvelle intégration à l'aide de la méthode des moindre carrés.

Programmation

Nous donnerons pour finir le programme Fortran qui aura servi à toutes nos intégrations, son mode de fonctionnement et ses principales caractéristiques.

PROBLEME DES N CORPS

Soient donc N corps P_j, $j = \left[0,1,2,\ldots,(N-1)\right]$, de masses respectives $m_0, m_1,\ldots m_{(N-1)}$.

Les axes de référence étant centrés en P_0, nous étudions les mouvements des P_i, $i = \left[1,2\ldots,(N-1)\right]$ autour de P_0.

Soient $\vec{r}_i = (x_i, y_i, z_i)$ leurs coordonnées, les équations différentielles sont données par :

$$\vec{r}_i^{(2)} = - K(m_0 + m_i) \frac{\vec{r}_i}{R_i^3} + \sum_{\substack{j=1 \\ j\neq i}}^{N-1} Km_j \left[\frac{(\vec{r}_j - \vec{r}_i)}{\Delta_{j,i}^3} - \frac{\vec{r}_j}{R_j^3} \right] \qquad (1)$$

avec : $\Delta_{j,i} = \left[(x_j - x_i)^2 + (y_j - y_i)^2 + (z_j - z_i)^2 \right]^{\frac{1}{2}}$

Définissons maintenant $3 \frac{(N-2)(N-1)}{2}$ variables auxiliaires et leurs dérivées par :

$$(\vec{r}_q^{(k)} - \vec{r}_p^{(k)}) = \vec{r}_l^{(k)} \quad l,\ p \text{ et } q \text{ entiers} > 0\ ;\ q > p \qquad (2)$$

telles que pour p variant dans l'intervalle $\left[1,(N-2)\right]$, (q variant alors de $(p+1)$ à $(N-1)$ quand p est fixé), on ait:

$$l = (N-1)+(N-2)+\ldots\ldots..(N-p)+(q-p) \qquad (3)$$

$$l=\left[N\ ,\ (N+1),\ldots\ldots,N'\right] \quad \text{avec} \quad N' = (N-1) + \frac{(N-2)(N-1)}{2}$$

Nous pourrons alors écrire (1) sous la forme générale suivante:

$$\vec{r}_i^{(2)} = - \sum_{n=1}^{N'} \mu_{i,n} \frac{\vec{r}_n}{R_n^3} \qquad (4)$$

$$n = \left[1,2\ldots..N'\right]$$

Les constantes $\mu_{i,n}$ constituent une matrice de i lignes et N' colonnes dans laquelle et sur une même ligne i on a :

<u>Pour n variant dans l'intervalle</u> $\left[1,(N-1)\right]$:

$\mu_{i,n} = K\, m_i$ sauf si $n = i$ auquel cas $\mu_{i,n} = \mu_{i,i} = K(m_0 + m_i)$

<u>Pour n variant dans l'intervalle</u> $\left[N,N'\right]$:

si $j > i$ $\quad \vec{r}_n = (\vec{r}_j - \vec{r}_i)$ et $\mu_{i,n} = -K\, m_j$

$$n = (N-1) + (N-2) + \ldots\ldots + (N-i) + (j-i)$$

si $j < i$ $\quad \vec{r}_n = (\vec{r}_i - \vec{r}_j)$ et $\mu_{i,n} = +K\, m_j$

$$n = (N-1) + (N-2) + \ldots\ldots + (N-j) + (i-j)$$

Tous les autres éléments d'une même ligne i sont nuls.

DERIVEES DES COORDONNEES

En dérivant k fois Les équations (4), nous obtenons :

$$\vec{r}_i^{\,(k+2)} = \frac{d^k}{dt^k}\left[- \sum_{n=1}^{N'} \mu_{i,n} \frac{\vec{r}_n}{R_n^3} \right] \tag{5}$$

On voit donc que si on peut connaitre formellement les dérivées successives de fonctions telles que : $F_{\vec{r}} = \dfrac{\vec{r}}{R^3}$ (6)

nous pourrons ensuite calculer celles des coordonnées $\vec{r}_i$. Dérivons donc les équations (6) et pour cela posons $G = \dfrac{1}{R^3}$, nous aurons alors $F_{\vec{r}} = G.\vec{r}$ (7)

Pour commencer nous calculerons les dérivées de R, puis à partir de celles ci les dérivées de G et pour finir celles des fonctions $F_{\vec{r}}$.

Nous disposerons pour cela des formules de récurrence suivantes, que l'on construit facilement.

Pour les fonctions R :

Si k = 1 :

$$R^{(1)} = \frac{1}{R}\left[\vec{r}.\,\vec{r}^{\,(1)}\right] \tag{8}$$

et pour k ≥ 2 :

$$R^{(k)} = \frac{1}{R}\left[\vec{r}.\vec{r}^{\,(k)} + \sum_{h=0}^{k-2} C_{k-1}^{h}\left[\vec{r}^{\,(k-1-h)}.\vec{r}^{\,(h+1)} - R^{(k-1-h)}.R^{(h+1)}\right]\right] \tag{9}$$

Pour les fonctions G :

si k = 1 :

$$G^{(1)} = -\,3\,G\,\frac{R^{(1)}}{R} \tag{10}$$

et pour k ≥ 2 :

$$G^{(k)} = -\frac{1}{R}\left[3\,G\,R^{(k)} + \sum_{h=0}^{k-2} C_{k-1}^{h}\left[R^{(k-1-h)}.G^{(h+1)} + 3\,G^{(k-1-h)}.R^{(h+1)}\right]\right] \tag{11}$$

Pour les fonctions $F_{\vec{r}}$:

Nous obtiendrons les dérivées $F_{\vec{r}}^{(k)}$ par :

$$F_{\vec{r}}^{(k)} = \sum_{h=0}^{k} C_k^h \, G^{(h)} \, \vec{r}^{(k-h)} \tag{12}$$

et nous aurons finalement celles des coordonnées $\vec{r}_i^{(k+2)}$ par :

$$\vec{r}_i^{(k+2)} = - \sum_{n=1}^{N'} \mu_{i,n} \, F_{\vec{r}_n}^{(k)} \tag{13}$$

SOLUTION

Les conditions initiales $\vec{r}_i(t_0)$, $\vec{r}_i(t_0)$ étant données, on en déduit les $\vec{r}_l(t_0)$, et les $\dot{\vec{r}}_l(t_0)$, $l = [N,N']$ en faisant $k = 0$ puis $k = 1$ dans les formules (2).

Connaissant alors les $\vec{r}_n(t_0)$ on calcule tous les $R(t_0)$,$G(t_0)$,et $F_{\vec{r}_n}(t_0)$,d'où les $\vec{r}_i^{(2)}(t_0)$ par les formules (13) et ensuite les $\vec{r}_l^{(2)}(t_0)$ par (2).

Connaissant aussi les $\dot{\vec{r}}_n(t_0)$, les formules (8), (10), (12) nous donnent respectivement les $R_n^{(1)}(t_0)$, $G_n^{(1)}(t_0)$, $F_{\vec{r}_n}^{(1)}(t_0)$,d'où les $\vec{r}_i^{(3)}(t_0)$ par (13) et les $\vec{r}_l^{(3)}(t_0)$ par (2).

Disposant alors des $\vec{r}_n^{(2)}(t_0)$,les formules (9), (11),et (12) nous fournirons les $R^{(2)}(t_0)$, $G^{(2)}(t_0)$, $F_{\vec{r}_n}^{(2)}(t_0)$ d'où les $\vec{r}_i^{(4)}$ et $\vec{r}_l^{(4)}$ par (13) et (2).......etc.

Nous calculerons ainsi de proche en proche tous les $\vec{r}_i^{(k)}(t_0)$ jusqu'à l'ordre k inclus; nous obtiendrons ensuite les $R_n^{(k)}(t_0)$ $G_n^{(k)}(t_0)$, et $F_{\vec{r}_n}^{(k)}(t_0)$ d'où la solution recherchée:

$$\vec{r}_i(t) = \sum_{h=0}^{k+2} \vec{r}_i^{(h)}(t_0) \frac{t^h}{h!} \tag{14}$$

INTEGRALES PREMIERES

Ecrivons les intégrales premières du problème des N corps sous une forme plus facile à utiliser par la suite.
Soit par exemple l' intégrale de l'énergie, elle s'écrit, M étant la somme de toutes les masses, (Tisserand,1896):

$$\sum_{i=1}^{N-1} m_i\,(\dot{x}_i^2 + \dot{y}_i^2 + \dot{z}_i^2)$$

$$-\frac{1}{M}\left[\left(\sum_{i=1}^{N-1} m_i\,\dot{x}_i\right)^2 + \left(\sum_{i=1}^{N-1} m_i\,\dot{y}_i\right)^2 + \left(\sum_{i=1}^{N-1} m_i\,\dot{z}_i\right)^2\right]$$

$$-\ 2\ K \sum_{i=1}^{N-1} \frac{m_0\ m_i}{R_i} \quad -\ 2\ K \sum_{j=1}^{N-2}\ \sum_{i=j+1}^{N-1} \frac{m_j\ m_i}{\Delta_{j,i}} = \text{constante} \qquad (15)$$

En multipliant l'équation (15) par M, et en nous servant des variables auxiliaires $\vec{r}_l$ et de leurs dérivées premières $\dot{\vec{r}}_l$ précédemment définies par les formules (2), i>j, on pourra écrire (15) après quelques calculs :

$$\sum_{i=1}^{N-1} m_0\ m_i (\dot{x}_i^2 + \dot{y}_i^2 + \dot{z}_i^2) + \sum_{j=1}^{N-2}\ \sum_{i=j+1}^{N-1} m_j\ m_i\ (\dot{x}_l^2 + \dot{y}_l^2 + \dot{z}_l^2)$$

$$-2\ K\ M \left[\sum_{i=1}^{N-1} \frac{m_0\ m_i}{R_i} + \sum_{j=1}^{N-2}\ \sum_{i=j+1}^{N-1} \frac{m_j\ m_i}{R_l}\right] = \text{constante} \qquad (16)$$

En convenant ensuite de remplacer l par i lorsque j = 0, l'intégrale de l'énergie s'écrira finalement :

$$2 \sum_{j=0}^{N-2}\ \sum_{i=j+1}^{N-1} m_j\ m_i\ (\ T_l\ -\ U_l\ ) = H = \text{constante} \qquad (17)$$

avec : $\quad 2\ T_l = (\dot{x}_l^2 + \dot{y}_l^2 + \dot{z}_l^2) \quad$ et $\quad U_l = \dfrac{K\ M}{R_l}$

De même pour les intégrales des aires on aura :

$$\sum_{j=0}^{N-2}\ \sum_{i=j+1}^{N-1} m_j\ m_i\ (\ \vec{r}_l \wedge \dot{\vec{r}}_l) = \text{constantes} \qquad (18)$$

CONDITIONS INITIALES

Pour obtenir les conditions initiales nous sommes partis d'une intégration numérique complète, l'éphéméride DE200 du JPL.
Celle ci étant ensuite placée dans l'écliptique dynamique et équinoxe J2000 par deux rotations successives, (BRETAGNON, 1982), on calcule tous les 20 jours ,entre le 25 Septembre 1891 à 0h TU (JJ=241200.5) et le 26 Février 2000 à 0h TU (JJ=2451600.5), les éléments osculateurs des planètes (a,e,i,Ω,ω,l) puis les quantités k,h,q,p, qui s'en déduisent :

$$k = e \cos(\Omega+\omega); \qquad h = e \sin(\Omega+\omega)$$

$$q = \sin \frac{i}{2} \cos \Omega; \; p = \sin \frac{i}{2} \sin \Omega$$

Ces quantités et a,l, sont alors corrigées de leurs perturbations relativistes,(BRETAGNON et LESTRADE,1982), et de celles dues à la Lune et aux petites planètes.
On dispose ainsi d'une éphéméride newtonienne de référence (E_R), à laquelle nous pourrons ultérieurement comparer nos résultats.
Les conditions initiales $\vec{r}_i(t_0)$ et $\dot{\vec{r}}_i(t_0)$ ont été calculées à partir de cette éphéméride, prise à la date du 26 Février 2000 à 0h TU (JJ=2451600.5).
Elles sont données dans le tableau 1, (BRETAGNON,1986),ainsi que la valeur de l'intégrale de l'énergie calculée par (17).
Le tableau 2 nous fournit ensuite les inverses des masses des planètes, celle du Soleil étant prise pour unité.Ces masses sont issues de DE200 (STANDISH,1982).

EPHEMERIDE

Partant des conditions initiales du tableau 1 , nous avons effectué une première intégration sur 40000 jours du système newtonien composé des neuf planètes principales dans leurs mouvements relatifs autour du Soleil.

Pour cela, nous avons programmé la méthode que nous venons de décrire et utilisé un pas d'intégration de -4 jours, nécessitant alors des développements en séries de Taylor comportant 25 dérivées.

Cette intégration a été réalisée sur le NAS 9080 du Centre Inter Regional de Calcul Electronique du CNRS et a duré environ 8 min et 20 sec.

Précision de la méthode

Afin d'évaluer la précision de la méthode utilisée, nous avons réintégré notre système, cette fois en sens inverse, en partant des conditions initiales (JJ=2411600.5) de la fin de notre première éphéméride. On soustrait alors des coordonnées et vitesses données dans le tableau 1, celles obtenues à la fin de cette nouvelle intégration (JJ=2451600.5).

Le tableau 3 rassemble l'ensemble de ces différences pour toutes les planètes.

On y constate tout d'abord que la dérive globale des résultats au bout de 80000 jours reste faible (incertitude relative sur l'intégrale de l'énergie = 3.10^{-12}), et qu'ensuite, les précisions relatives (coordonnées / demi grands axes) sont inférieures à $1,2.10^{-9}$(abscisse de Vénus) pour les planètes inférieures, et à $1,15.10^{-10}$(ordonnée de Jupiter) pour les planètes supérieures.

Comparaison avec DE200 (JPL)

Afin de comparer les résultats de notre première intégration avec ceux de l'éphéméride de référence (E_R) précédemment définie, on commence par calculer, tous les 20 jours, les élements osculateurs des planètes, puis les quantités k,h,q,p qui s'en déduisent. On obtient ainsi une éphéméride (E_1) que l'on peut soustraire à (E_R)

On traite ensuite les différences obtenues par la méthode des moindres carrés, afin d'apporter de légères modifications à chacun des a,e,l,k,h,q,p de départ de chaque planète.

On peut alors calculer de nouvelles coordonnées et vitesses initiales pour une autre intégration, d'où une nouvelle éphéméride (E_2) plus proche de (E_R) que (E_1).

On procédera ainsi jusqu'à ce que les corrections à apporter aux conditions initiales deviennent négligeables pour la précision demandée.

Le tableau 4 nous donne les coordonnées et vitesses initiales obtenues de cette façon, le tableau 5 celles, finales, de l'éphéméride construite par une dernière intégration de 40000 jours.

Celle ci, comparée à DE200 nous a fourni des différences d'une précision d' environ 10^{-10} UA sur les coordonnées rectangulaires.

CONCLUSION

Grâce à des développements en séries de Taylor ayant un degré élevé par rapport aux dérivées, nous avons pu conserver un pas d'intégration plus grand que ceux généralement utilisés dans les méthodes usuelles d'intégration numérique.(aux voisinages des périastres notamment).

De ce fait, et grâce aussi à l'analycité des formules, les erreurs d'arrondi s'en sont trouvé grandement diminuées.

Ainsi nous avons pu observer une bonne précision des résultats sur une longue période de temps, et ceci malgré la présence de Mercure dans le système que nous avons pris pour exemple.

Au vu du bon comportement de l'intégrale de l'énergie nous pourrions d'ailleurs agrandir encore plus l'intervalle d'intégration.

Notons enfin que cette méthode, dont la programmation est facile à mettre en oeuvre, permet l'étude de n'importe quel système newtonien (comète par exemple), à condition d'en exclure l'éventualité d'un choc entre deux ou plusieurs corps.

TABLEAU 1. POSITIONS(UA) ET VITESSES(UA/J) DES PLANETES
A L'INSTANT INITIAL D'INTEGRATION
PLAN DE REFERENCE: ECLIPTIQUE DYNAMIQUE ET EQUINOXE J 2000
JOUR JULIEN = 2451600.5

PLANETES	POSITIONS	VITESSES
MERCURE	-0.2503322266997050	-0.0243880740618594
	0.2219933134940310	-0.0199006661918177
	0.0411116039958294	0.0006127790621238
VENUS	0.0174780967869107	0.0200854700002338
	-0.7267426358190740	0.0004115973599546
	-0.0109417141664655	-0.0011537145705727
TERRE-LUNE	-0.9091915624774220	-0.0070858429584318
	0.3916073804088610	-0.0158655216295262
	0.0000006337357788	0.0000000304051933
MARS	1.2030185340340200	-0.0071244586662406
	0.7867887615327550	0.0129051790708794
	-0.0130904757456262	0.0004454747030769
JUPITER	3.7330739975155200	-0.0050865465611740
	3.2848131838485899	0.0060262786120076
	-0.0972162197545890	0.0000889035628590
SATURNE	6.1644217605744700	-0.0044268306574236
	6.7819637930593399	0.0037473494696568
	-0.3631445050232320	0.0001107931782759
URANUS	14.5796795804070001	0.0026475039005944
	-13.5852040761606001	0.0027006548067743
	-0.2395026654553700	-0.0000242628753771
NEPTUNE	16.9549722100605003	0.0025686464544125
	-24.8926233810363016	0.0017915021721705
	0.1218939840518710	-0.0000959701485907
PLUTON	-9.7077482397720101	0.0030340544888208
	-28.0439936763418984	-0.0015216209707127
	5.8109566466828999	-0.0007156387594053

INTEGRALE DE L'ENERGIE : UNITE = 1.E-7

-0.6654158715721031

TABLEAU 2 . INVERSES DES MASSES DES PLANETES ISSUS DE L'EPHEMERIDE DE200 DU JPL

MASSE DU SOLEIL = 1

PLANETES	
MERCURE	6023600.00000
VENUS	408523.50000
TERRE-LUNE	328900.55000
MARS	3098710.00000
JUPITER	1047.350010905518
SATURNE	3498.00000
URANUS	22960.00000
NEPTUNE	19314.00000
PLUTON	130000000.00000

TABLEAU 3. DIFFERENCES ENTRE POSITIONS(UA) ET VITESSES(UA/J) DES PLANETES APRES INTEGRATION ALLER RETOUR(80000 J)

JOUR JULIEN = 2451600.5

UNITE = 1.E-10

PLANETES	POSITIONS	VITESSES
MERCURE	2.80	-0.22
	2.29	0.20
	-0.07	0.04
VENUS	-8.66	0.01
	-0.19	-0.24
	0.50	0.00
TERRE-LUNE	3.97	-0.16
	8.93	0.07
	0.00	0.00
MARS	8.81	0.15
	-15.83	0.10
	-0.55	0.00
JUPITER	5.26	0.01
	-5.98	0.01
	-0.09	0.00
SATURNE	4.47	0.00
	-3.38	0.00
	-0.12	0.00
URANUS	-1.88	0.00
	-2.87	0.00
	0.01	0.00
NEPTUNE	-1.32	0.00
	-2.06	0.00
	0.07	0.00
PLUTON	-1.50	0.00
	-0.08	0.00
	0.44	0.00

DIFFERENCE SUR L'INTEGRALE DE L'ENERGIE : UNITE = 1.E-17

0.02

TABLEAU 4. POSITIONS(UA) ET VITESSES(UA/J) DES PLANETES
APRES CORRECTIONS
PLAN DE REFERENCE: ECLIPTIQUE DYNAMIQUE ET EQUINOXE J 2000
JOUR JULIEN = 2451600.5

PLANETES	POSITIONS	VITESSES
MERCURE	-0.2503322272474241	-0.0243880740187149
	0.2219933130543761	-0.0199006662291828
	0.0411116040134522	0.0006127790550368
VENUS	0.0174780969832841	0.0200854700008875
	-0.7267426357972082	0.0004115973655980
	-0.0109417141864230	-0.0011537145704358
TERRE-LUNE	-0.9091915626221172	-0.0070858429546628
	0.3916073801866026	-0.0158655216304405
	0.0000006337269945	0.0000000304053757
MARS	1.2030185338987920	-0.0071244586683808
	0.7867887618949421	0.0129051790684479
	-0.0130904754207521	0.0004454747041100
JUPITER	3.7330739970653830	-0.0050865465617725
	3.2848131843439790	0.0060262786111905
	-0.0972162197513344	0.0000889035628536
SATURNE	6.1644217608860861	-0.0044268306570807
	6.7819637928089400	0.0037473494696738
	-0.3631445050315127	0.0001107931782389
URANUS	14.5796795810394000	0.0026475039006449
	-13.5852040756255601	0.0027006548067234
	-0.2395026654509072	-0.0000242628754072
NEPTUNE	16.9549722088488011	0.0025686464546824
	-24.8926233783199109	0.0017915021721450
	0.1218939840126886	-0.0000959701486240
PLUTON	-9.7077482547662419	0.0030340544879385
	-28.0439936839154491	-0.0015216209715392
	5.8109566518888340	-0.0007156387590526

INTEGRALE DE L'ENERGIE : UNITE = 1.E-7

-0.6654158716182916

TABLEAU 5. POSITIONS(UA) ET VITESSES(UA/J) DES PLANETES

A L'INSTANT FINAL D'INTEGRATION

PLAN DE REFERENCE: ECLIPTIQUE DYNAMIQUE ET EQUINOXE J 2000

JOUR JULIEN = 2411600.5

PLANETES	POSITIONS	VITESSES
MERCURE	-0.2576687518872368	0.0175868799212131
	-0.3791513241443215	-0.0144665374801825
	-0.0071950703512374	-0.0027988495000633
VENUS	-0.0478396027134483	0.0200455386154473
	-0.7251830666565513	-0.0014100355266644
	-0.0069246827363900	-0.0011776646939453
TERRE-LUNE	0.8703804807184863	0.0084796600188988
	-0.5149679396699509	0.0147442829692835
	-0.0001096032600599	0.0000038603797076
MARS	0.6362707422639920	0.0130220279289513
	-1.2551195580973074	0.0075226527499204
	-0.0420319593297117	-0.0001658371106017
JUPITER	3.3079184999203806	0.0056401087405234
	-3.8541443085480818	0.0052787064953480
	-0.0585602232678248	-0.0001480045741013
SATURNE	-8.6498328693881499	-0.0023396860855807
	3.4092364104726043	-0.0052006445435954
	0.2828278405758518	0.0001845459284167
URANUS	-16.3165791780625433	0.0018104636390494
	-8.6194029169052933	-0.0036590291030088
	0.1801147430157560	-0.0000372832508783
NEPTUNE	11.9722583886743688	-0.0028885982212621
	27.3074939138436470	0.0012836838050364
	-0.8380811427838262	0.0000401984022563
PLUTON	17.5090297675820956	-0.0020324972718968
	43.6921163840676954	0.0006366390838723
	-9.7426428668190121	0.0005213576991192

INTEGRALE DE L'ENERGIE : UNITE = 1.E-7

-0.6654158716193928

REFERENCES

BRETAGNON,P.:1982, Astron.Astrophys.114,278

BRETAGNON,P.:1986, CELEST.MECH.38,181

BROUCKE,R.:1971, CELEST.MECH. 4,110

LESTRADE,J.- F. et BRETAGNON,P. :1982, Astron.Astrophys.105,42

STANDISH,E.- M.:1982, Astron.Astrophys.114,297

STANDISH,E.- M. : 1982, DE200,Magnetic Tape.

TISSERAND,F.:1896, TRAITE de MECANIQUE CELESTE I, Gauthier-Villars, Paris

PROGRAMME D'INTEGRATION NUMERIQUE
(Problème des N Corps)

Le programme Fortran que nous avons utilisé se compose essentiellement d'une partie INITIALISATION et d'une partie INTEGRATION auxquelles se rattachent cinq SOUS PROGRAMMES et éventuellement un sixieme si on veut aussi stocker les résultats sur disques.

Pour commencer le programme fournit les paramètres de l'intégration

1) le nombre de corps étudiés par rapport au corps central:

$$NC=N-1$$

2) le nombre de dérivées des coordonnées nécessaires à chaque série de Taylor utilisée :

$$ND$$

3) le tiers du nombre total de variables utilisées :

$$NP = N' = \frac{N(N-1)}{2}$$

soit $(N-1)$ variables naturelles et $\frac{(N-1)(N-2)}{2}$ variables auxiliaires.

Par lecture d'un fichier on obtient ensuite:

1) le temps initial TINIT, le temps final TFIN, le pas d'intégration PAINIT, le pas d'impression NIMP, éventuellement le pas de stockage sur disque NENR, la précision demandée PRECR, l'inverse de la masse du corps central AMAS(0).

2) les inverses des masses qui sont lues dans le tableau AMAS, les conditions initiales $\vec{r}_i(t_0)$ et $\dot{\vec{r}}_i(t_0)$ de l'intégration $i = [1, N-1]$, qui sont lues dans le tableau V0.

INITIALISATION

Cette partie du programme effectue une fois pour toutes certains calculs essentiels pour la suite.

Au préalable on y calcule les masses dans AMAS et leur somme dans SUM multipliées par la constante de Gauss.

On calcule ensuite les coéfficients numériques des formules de récurrence qui permettront d'obtenir les dérivées successives des fonctions R et $F_{\vec{r}}$ d'une part et celles des fonctions G d'autre part. Ils seront donnés dans les tableaux CR et CG.

On y construit pour finir la matrice composée des $\mu_{i,n}$,c'est le tableau AMU.

Tableaux CR et CG

Ce sont des tableaux à deux indices dont le premier indique l'ordre de dérivation et le second le rang du terme que l'on considère dans le développement correspondant de cette dérivation.

Comme celles ci seront toutes de la forme $\dfrac{d^k(ab)}{dt^k}$

les coefficients numériques de ces termes sont ceux du développement du binome ou en sont directement issus.

1) CR

Les éléments du tableau CR serviront à la fois aux calculs des dérivées successives des fonctions R et à celles des fonctions $F_{\vec{r}}$.

On commence par écrire le tableau CR comme étant le triangle de Pascal, puis pour tenir compte des symétries qui apparaîtront ultérieurement dans les calculs, on divise par 2 le coefficient central de chacune de ses lignes quand l'ordre de dérivation est pair.

2) CG

Les éléments du tableau CG sont obtenus de proche en proche à partir de l'équation formelle :

$$RG^{(1)} = - 3GR^{(1)}$$

dont on dérive successivement les deux membres suivant la règle de Leibnitz.

A chaque nouvelle dérivation on regroupe alors les termes semblables et on obtient ainsi les coefficients numériques recherchés. En les multipliant ensuite par -1 pour les rendre positifs, on a le tableau CG.

Tableau AMU

Ce tableau est la matrice qui servira aux calculs des $\vec{r}_i^{(k)}$, ses éléments en sont les $\mu_{i,n}$:

$$\vec{r}_i^{(k)} = - \sum_{n=1}^{N'} \mu_{i,n} \; F_{\vec{r}_n}^{(k-2)} \qquad n = [1,N']$$

k étant un indice pouvant varier de 0 à ND, avec ici $k = [2,ND]$

INTEGRATION

Les coordonnées et vitesses à l'instant $t = t_0$ étant dans le tableau V_0, on les transfère dans le tableau V.

En appelant le sous programme VD (k=0,1) on obtient ensuite les valeurs des variables auxiliaires et de leurs dérivées premières.

On peut alors calculer tous les R_n, G_n et $F_{\vec{r}_n}$ $n=[1,N']$,

et appeler ensuite le sous programme STOC (k=2), qui calcule tous les $\vec{r}_n^{(2)}(t)$ et les stocke dans V.

Avant de commencer une première ou une nouvelle intégration, on appelle le sous programme IMPRIM qui calcule les intégrales premières et les imprime à la suite des coordonnées et vitesses des corps étudiés.

On examine aussi la grandeur du pas utilisé en le comparant à celui initialement prévu et conservé dans PA; suivant les cas on effectuera l'intégration avec un pas divisé ou celui d'origine.

Enfin l'instruction 50 fait avancer le temps, le programme s'arrête si TEMPS > TFIN.

S'il ne l'est pas on procéde alors aux calculs des dérivées.

Calculs des dérivées

I est l'ordre de dérivation des fonctions $F_{\vec{r}}$; il varie de 0 jusqu'à ND-2; les dérivées des coordonnées seront donc connues jusqu'à l' ordre ND inclus.

On utilise pour cela les formules de récurrence que nous avons construites et qui permettent d'obtenir de proche en proche les dérivées $I^{\text{éme}}$ des R_n (Formules 8 et 9), puis celles des G_n (Formules 10 et 11) et enfin celles des $F_{\vec{r}_n}$ (Formule 12).

A la fin de chaque étape I on appelle le sous programme STOC (k=I+2), qui connaissant les $F_{\vec{r}_n}^{(I)}$ calcule les dérivées $\vec{r}_n^{(k)}$ et les stocke dans V.

On revient ensuite si besoin est aux calculs des dérivées d'un ordre supérieur, jusqu'à atteindre l'ordre I=ND-2.

Remarque

L est un indice qui varie de 1 à I/2 , ce qui fait que le nombre des coéfficients du tableau CR utilisés pour deux dérivations successives (impaire et paire) reste le même.

SOUS PROGRAMMES

Il y a six sous programmes dont voici pour chacun leur spécificité:

TN

Ce sous programme calcule les $T_k = \frac{(pas)^k}{k!}$ pour k variant de 0 jusqu'a ND.

VD

Connaissant les valeurs des dérivées $\vec{r}_i^{(k)}$, ce sous programme en déduit celles des variables auxiliaires $\vec{r}_l$ soit $\vec{r}_l^{(k)}$:

$$(\vec{r}_q^{(k)} - \vec{r}_p^{(k)}) = \vec{r}_l^{(k)} \qquad k = \left[\ 0,ND\ \right]\ ,\ l = \left[\ N,N'\right]$$

(Formules 2)

STOC

C'est le plus important des sous programmes; il comporte plusieurs parties.

1) Calculs et Stockages des dérivées des coordonnées

Connaissant les dérivées $I^{\text{èmes}}$ des $F_{\vec{r}_n}$ il calcule pour commencer les valeurs des dérivées $\vec{r}_i^{(k)}$ (k=I+2) avec l'aide de la matrice AMU; il obtient ensuite celles des variables auxiliaires en appelant le sous programme VD. Les résultats sont stockés dans le tableau V , puis :

si k < ND on retourne à une dérivation supérieure (RETURN 0)

si k = ND on procéde à un test de précision.

2) Test

Lorsque k = ND il consiste à faire la somme TST des valeurs absolues de toutes les dérivées $\vec{r}_i^{(k)}$, à la multiplier par la valeur absolue de T_k, et à la comparer à PMIN.

PMIN = PRECR multipliée par le minimum des distances mutuelles entre tous les corps du système, à l'instant que l'on considère.

3) Résultats

Si la somme TST est inférieure à la précision requise PMIN,on calcule les séries de Taylor résultats dans VO et on imprime ensuite ces résultats. (RETURN 1 renvoie à l'instruction 5000)

4) Changement de pas

Si la somme TST est supérieure à PMIN on divise le pas par 2 et on retourne au test précédent aprés avoir appelé le sous programme TN qui recalcule les nouveaux T_k. On procéde ainsi jusqu'à ce que TST devienne inférieure à PMIN et on calcule alors les résultats.

Les intégrations suivantes se feront avec ce nouveau pas (s'il n'est pas de nouveau divisé) jusqu'à atteindre l'extrémité de l'intervalle initialement prévue. On reprendra alors le pas d'intégration initial PAINIT pour poursuivre l'intégration s'il y a lieu.

IMPRIM

Connaissant les $\vec{r}_i$ et les $\dot{\vec{r}}_i$ ainsi que les R_i et les G_i à un instant donné t, il appelle INTP.

Il imprime ensuite le jour julien correspondant à cet instant, puis les coordonnées et les vitesses des corps étudiés ainsi que les intégrales premières .

INTP

Il fournit les valeurs des quatre intégrales premières du problème des N Corps en mouvement relatif calculées à l'aide des formules 17 et 18.

ENREG

Ce sous programme peut être rajouté si l'on désire aussi stocker les résultats sur disques.

CONTROLE du PROGRAMME

Un pas d'intégration étant fixé à priori, nous avons alors à déterminer le nombre optimal de dérivées qui seront utilisées dans chaque série de Taylor du programme.

Le problème consiste à rendre la dérive de ce programme la plus petite possible pour un temps d'intégration minimum.

En effet si le nombre de dérivées employées dans les séries de Taylor est insuffisant le programme aura alors recours à des changements de pas plus ou moins nombreux ce qui augmentera les erreurs d'arrondi.

Si au contraire le nombre des dérivées est trop important cela augmentera le temps d'intégration sans pour autant apporter de précision supplémentaire.

Pour éviter ces deux inconvénients on procéde de la manière suivante :

On se base sur le corps le plus rapide du système que l'on étudie, Mercure par exemple pour le système solaire, et on calcule ses coordonnées et vitesses au bout de l'une de ses révolutions avec un nombre de dérivées (insuffisant) que l'on se donne à l'avance.

On recommencera plusieurs fois la même intégration mais en introduisant à chaque nouvelle fois une dérivée supplémentaire dans les séries, jusqu'à ce que les différences constatées sur les coordonnées et les vitesses entre deux intégrations successives deviennent négligeables pour la précision requise.

On pourra ainsi déterminer le nombre optimal ND des dérivées le mieux adapté au programme.

Si ce nombre ND est suffisamment élevé il permettra aussi de conserver un pas pratiquement constant durant toute la durée de l'intégration.

```
************************************************************************
*         PROGRAMME FORTRAN D'INTEGRATION NUMERIQUE DE N CORPS         *
************************************************************************

      IMPLICIT DOUBLE PRECISION (A-H , O-Z)

      PARAMETER (NC=9, ND=25,NP=NC*(NC+1)/2)
      DIMENSION VO(3,NC,0:1),AMAS(0:NC),AMU(NC,NP),CR(ND,ND),CG(ND,ND)
      DIMENSION V(3,NP,0:ND),R(NP,0:ND),G(NP,0:ND),F(3,NP),T(0:ND)
      DIMENSION TP(NP),UP(NP),CP(3,NP)

      GK=0.01720209895D0

      G2=GK*GK

      READ *, TINIT,TFIN,PAINIT,NIMP,NENR,PRECR
      PRINT *, 'TINIT = ',TINIT,' TFIN = ',TFIN,' PAS D''INTEGRATION =',
     1 PAINIT
      PRINT *, ' PAS D''IMPRESSION = ',NIMP,' PAS DE STOCKAGE SUR DISQUE
     1 =',NENR,'PRECISION : ',PRECR
      READ *, AMAS(0)
      PRINT *, 'INVERSE DE LA MASSE DU CORPS CENTRAL :',AMAS(0)

      DO 1 N=1,NC
      READ *, AMAS(N)
      READ *, (VO(K,N,0),K=1,3)
      READ *, (VO(K,N,1),K=1,3)
      PRINT 1000, N,AMAS(N),(VO(K,N,0),K=1,3)
      PRINT 1001,          (VO(K,N,1),K=1,3)
1000  FORMAT(1X,I4,F18.4,3(D30.16))
1001  FORMAT(23X,3(D30.16))
1     CONTINUE
```

```
*
*********************   INITIALISATION   **********************
*
      DO 2 N = 0,NC
2     AMAS(N) = G2/AMAS(N)
      SUM=0
      DO 3 N=0,NC
3     SUM=SUM+AMAS(N)
      CR(1,1)=1
      CR(1,2)=1
      DO 10 I=2,ND-1
      IH=I+1
      J=I-1
      CR(I,1)=1
      CR(I,IH)=1
      DO 11 K=2,I
11    CR(I,K)=CR(J,K)+CR(J,K-1)
10    CONTINUE
      DO 12 I=2,ND-2,2
      J=I/2+1
12    CR(I,J)=CR(I,J)/2
      CG(1,1)=3
      CG(2,1)=3
      CG(2,2)=4
      DO 13 I=3,ND-1
      CG(I,1)=3
      CG(I,I)=I+2
      J=I-1
      DO 14 K=2,J
14    CG(I,K)=CG(J,K)+CG(J,K-1)
13    CONTINUE
      DO 15 I=1,NC
      DO 15 J=1,NC
15    AMU(I,J)=AMAS(J)
      DO 16 I=1,NC
16    AMU(I,I)=AMAS(0)+AMAS(I)
```

```
      JMIN=NC+1
      JD=NC-1
      DO 18 I=1,NC-1
      L=JMIN-I-1
      JD=JD-1
      JMAX=JMIN+JD
      DO 19 J=JMIN,JMAX
19    AMU(I,J)=-AMAS(J-L)
18    JMIN=JMAX+1
      M=NC
      DO 20 J=1,NC-1
      L=NC-J
      DO 21 I=1,L
21    AMU(J+I,M+I)=AMAS(J)
20    M=M+L
```

```
*
********************    INTEGRATION    ********************
*
      PAS=PAINIT
      TINT=TINIT
      PA=PAS
      PAT=0
      TEMPS=TINIT
      CALL TN(T,ND,PAS)
5000  CONTINUE
      DO 30 I=0,ND
      DO 30 J=1,NP
      R(J,I)=0
      G(J,I)=0
      DO 31 K=1,3
31    V(K,J,I)=0
30    CONTINUE
      DO 25 K=0,1
      DO 25 J=1,NC
      DO 25 I=1,3
25    V(I,J,K)=V0(I,J,K)
      DO 26 K=0,1
      CALL VD(K,NC,ND,NP,V)
26    CONTINUE
      RMIN=1.D50
      DO 27 J=1,NP
      R(J,0)=DSQRT(V(1,J,0)**2+V(2,J,0)**2+V(3,J,0)**2)
      G(J,0)=1/R(J,0)**3
      RMIN=DMIN1(RMIN,R(J,0))
      DO 28 I=1,3
28    F(I,J)=G(J,0)*V(I,J,0)
27    CONTINUE
      PMIN=RMIN*PRECR
      CALL STOC(2,NC,ND,NP,V0,V,AMU,F,T,TEMPS,PAS,PMIN,*5000)
      BPAS=(TEMPS-TINIT)/PAINIT
      TBI=MOD(BPAS/NIMP,1.D0)
      IF(TBI.LE.1.D-6.OR.TBI.GE.1.D0-1.D-6)
     1         CALL IMPRIM(NC,ND,NP,V0,V,AMAS,SUM,TP,UP,CP,R,TEMPS,G2)
```

```
      IF(PAS.EQ.PA) GO TO 50
      PAT=PAT+PAS
      IF(DABS(PA-PAT).GT.1.D-10) GO TO 50
      PAT=0
      PAS=PA
      CALL TN(T,ND,PAS)
50    TEMPS=TEMPS+PAS
************************************************************************
*     WRITE(6,7001) TEMPS,TFIN
7001  FORMAT(1X,'TEMPS = ',F12.2,'  TFIN = ',F12.2)
************************************************************************
      IF(TEMPS*PAS.GT.TFIN*PAS) STOP
      DO 40 I=1,ND-2
      K=I+2
      IP=I/2
      IQ=I-1
      DO 41 J=1,NP
      DO 41 KP=1,3
41    F(KP,J)=0
      DO 42 J=1,NP
      W=1/R(J,0)
      WX=V(1,J,0)
      WY=V(2,J,0)
      WZ=V(3,J,0)
      DO 43 L=1,IP
      IH=I-L
      M=L+1
      R(J,I)=R(J,I)+CR(I,M)*(V(1,J,L)*V(1,J,IH)+V(2,J,L)*V(2,J,IH)+
     1                       V(3,J,L)*V(3,J,IH)-R(J,L)*R(J,IH))
43    CONTINUE
      R(J,I)=(R(J,I)+WX*V(1,J,I)+WY*V(2,J,I)+WZ*V(3,J,I))*W
      DO 44 L=0,IQ
      IH=I-L
      M=L+1
      G(J,I)=G(J,I)+CG(I,M)*G(J,L)*R(J,IH)
44    CONTINUE
      G(J,I)=-G(J,I)*W
```

```
      DO 45 L=0,IP
      IH=I-L
      M=L+1
      W1=CR(I,M)*G(J,IH)
      W2=CR(I,M)*G(J,L)
      DO 46 KP=1,3
      F(KP,J)=F(KP,J)+W1*V(KP,J,L)+W2*V(KP,J,IH)
46    CONTINUE
45    CONTINUE
42    CONTINUE
      CALL STOC(K,NC,ND,NP,V0,V,AMU,F,T,TEMPS,PAS,PMIN,*5000)
40    CONTINUE
      STOP 4040
      END
```

```
      SUBROUTINE TN(T,ND,PAS)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION T(0:ND)
      T(0)=1
      DO 1 K=1,ND
      T(K)=T(K-1)*PAS/K
1     CONTINUE
      RETURN
      END
```

```
      SUBROUTINE VD(K,NC,ND,NP,V)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION V(3,NP,0:ND)
      JMIN=NC+1
      JD=NC-1
      DO 18 I=1,NC-1
      L=JMIN-I-1
      JD=JD-1
      JMAX=JMIN+JD
      DO 19 J=JMIN,JMAX
      DO 30 N=1,3
30    V(N,J,K)=V(N,J-L,K)-V(N,I,K)
19    CONTINUE
18    JMIN=JMAX+1
      RETURN
      END
```

```
      SUBROUTINE STOC(K,NC,ND,NP,V0,V,AMU,F,T,TEMPS,PAS,PMIN,*)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION V0(3,NC,0:1),V(3,NP,0:ND),AMU(NC,NP),F(3,NP),T(0:ND)
      DO 1 L=1,NC
      DO 1 J=1,NP
      DO 1 N=1,3
      V(N,L,K)=V(N,L,K)-AMU(L,J)*F(N,J)
1     CONTINUE
      CALL VD(K,NC,ND,NP,V)
      IF(K.LT.ND) RETURN 0
***************************    TEST    *************************
      TRT=0
      DO 2 L=1,NC
      DO 2 N=1,3
      TRT=TRT+DABS(V(N,L,K))
2     CONTINUE
10    CONTINUE
***********************************************************************
*     WRITE(6,7000) TRT,T(K),NC,ND,NP,TEMPS
7000  FORMAT(1X,'STOC',2D10.2,3I4,F14.2)
***********************************************************************
      TST=TRT*DABS(T(K))
      IF(TST.LT.PMIN) GO TO 20
CHANGEMENT DE PAS
      PAS=PAS/2
      TEMPS=TEMPS-PAS
      CALL TN(T,ND,PAS)
      GO TO 10
CE SONT LES RESULTATS
20    DO 4 N=1,NC
      DO 4 NR=1,3
      DO 4 J=0,1
      V0(NR,N,J)=0
      DO 4 KP=ND-J,0,-1
      V0(NR,N,J)=V0(NR,N,J)+V(NR,N,KP+J)*T(KP)
4     CONTINUE
      RETURN 1
      END
```

```
      SUBROUTINE IMPRIM(NC,ND,NP,V0,V,AMAS,SUM,TP,UP,CP,R,TEMPS,G2)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION V0(3,NC,0:1),AMAS(0:NC),TP(NP),UP(NP),CP(3,NP),AIA(3)
      DIMENSION R(NP,0:ND),V(3,NP,0:ND)
      CALL INTP(NC,ND,NP,AMAS,SUM,TP,UP,CP,AIA,AING,R,V,G2)
      WRITE(6,5010)
 5010 FORMAT(1H1)
      WRITE(6,1000) TEMPS
 1000 FORMAT(/,60X,F10.2)
      WRITE(6,2001)
 2001 FORMAT(/,14X,'X',20X,'Y',20X,'Z',20X,'XP',19X,'YP',19X,'ZP')
      WRITE(6,2000)
 2000 FORMAT(1X)
      DO 1 N=1,NC
      WRITE(6,1001) N,((V0(K,N,J),K=1,3),J=0,1)
 1001 FORMAT(1X,I2,6F21.16)
 1    CONTINUE
      WRITE(6,1002) AIA,AING
 1002 FORMAT(/,6X,4D25.16,/)
      RETURN
      END
```

```
      SUBROUTINE INTP(NC,ND,NP,AMAS,SUM,TP,UP,CP,AIA,AING,R,V,G2)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION AMAS(0:NC),TP(NP),UP(NP),CP(3,NP),AIA(3)
      DIMENSION R(NP,0:ND),V(3,NP,0:ND)
      DO 1 J=1,NP
      UP(J)=2*SUM/R(J,0)
      TP(J)=V(1,J,1)**2+V(2,J,1)**2+V(3,J,1)**2
      DO 2 K=1,3
      K2=K+1-3*(K/3)
      K3=K+2-3*(K/2)
2     CP(K,J)=V(K2,J,0)*V(K3,J,1)-V(K3,J,0)*V(K2,J,1)
1     CONTINUE
      DO 3 K=1,3
3     AIA(K)=0
      AING=0
      JMIN=1
      JD=NC
      DO 4 N=1,NC
      JD=JD-1
      JMAX=JMIN+JD
      DO 5 J=JMIN,JMAX
      I=J-JMIN+N
      DO 6 K=1,3
6     AIA(K)=AIA(K)+AMAS(N-1)*AMAS(I)*CP(K,J)
      AING=AING+AMAS(N-1)*AMAS(I)*(TP(J)-UP(J))
5     CONTINUE
      JMIN=JMAX+1
4     CONTINUE
      DO 7 K=1,3
7     AIA(K)=AIA(K)/G2**2
      AING=AING/G2**2
      RETURN
      END
```

```
      SUBROUTINE ENREG(NC,ND,NP,BPAS,NENR,TEMPS,V0,AMAS,SUM,TP,UP,
     1                 CP,R,V,G2)
      IMPLICIT DOUBLE PRECISION (A-H , O-Z)
      DIMENSION V0(3,NC,0:1),AIA(3)
      OPEN(40,ACCESS='DIRECT',RECL=(6*NC+5)*8)
      CALL INTP (NC,ND,NP,AMAS,SUM,TP,UP,CP,AIA,AING,R,V,G2)
      IENR=BPAS/NENR+1
      WRITE(40,REC=IENR) TEMPS,V0,AIA,AING
      RETURN
      END
```

```
2451600.5D0   2451500.5D0   -4.D0   25   5   1.D-16
1.D0
6023600.D0
-2.503322272474241D-01  2.219933130543761D-01  4.111160401345221D-02
-2.438807401871489D-02 -1.990066622918275D-02  6.127790550368159D-04
408523.5D0
 1.747809698328411D-02 -7.267426357972082D-01 -1.094171418642303D-02
 2.008547000088745D-02  4.115973655979915D-04 -1.153714570435756D-03
328900.55D0
-9.091915626221172D-01  3.916073801866026D-01  6.337269944815823D-07
-7.085842954662788D-03 -1.586552163044054D-02  3.040537566320542D-08
3098710.D0
 1.203018533898792D+00  7.867887618949421D-01 -1.309047542075205D-02
-7.124458668380764D-03  1.290517906844793D-02  4.454747041100437D-04
1047.350010905518D0
 3.733073997065383D+00  3.284813184343979D+00 -9.721621975133443D-02
-5.086546561772504D-03  6.026278611190509D-03  8.890356285362525D-05
3498.0D0
 6.164421760886086D+00  6.781963792808940D+00 -3.631445050315127D-01
-4.426830657080693D-03  3.747349469673750D-03  1.107931782389343D-04
22960.D0
 1.457967958103940D+01 -1.358520407562556D+01 -2.395026654509072D-01
 2.647503900644948D-03  2.700654806723400D-03 -2.426287540722319D-05
19314.D0
 1.695497220884880D+01 -2.489262337831991D+01  1.218939840126886D-01
 2.568646454682388D-03  1.791502172145005D-03 -9.597014862399412D-05
130000000.D0
-9.707748254766242D+00 -2.804399368391545D+01  5.810956651888834D+00
 3.034054487938522D-03 -1.521620971539244D-03 -7.156387590526466D-04
```